and generally the "n-th difference" is defined as

$$\Delta^n f_K = \sum_{j=0}^{n} (-1)^{n-1} \binom{n}{j} f_{K+j} \quad ,$$

where $\binom{n}{j} = \frac{n!}{j!\,(n-j)!}$ is binomial coefficient,

which can be found also from Pascal triangle:

$$\begin{array}{ccccccccccc}
 & & & & & 1 & & & & & \\
 & & & & 1 & & 1 & & & & \\
 & & & 1 & & 2 & & 1 & & & \\
 & & 1 & & 3 & & 3 & & 1 & & \\
 & 1 & & 4 & & 6 & & 4 & & 1 & \\
1 & & 5 & & \cdots & & & & & & \\
\cdot^{\cdot^{\cdot}} & & & & & & & & & &
\end{array}$$

Remark: example of 3-rd difference by Maple

```
with(LREtools):
delta(f(n),n,3);
```

90

Z- transform of 1-st difference

$$\mathcal{Z}\{\Delta^1 f_k\} = \sum_{k=0}^{\infty} \frac{f_{k+1}}{z^k} - \sum_{k=0}^{\infty} \frac{f_k}{z^k} =$$

$$= z\sum_{k=1}^{\infty} \frac{f_k}{z^k} - F(z) = z\sum_{k=0}^{\infty} \frac{f_k}{z^k} - z f_0 -$$

$$-F(z) = (z-1)F(z) - z f_0 \, ; \quad (*)$$

where $F(z) = \mathcal{Z}\{f_k\}$

and the transform of n-th difference is

$$\mathcal{Z}\{\Delta^n f_k\} = (z-1)^n F(z) - z\sum_{j=0}^{n-1} (z-1)^{n-1-j} \Delta^j f_0$$

Proof (by induction):

for $n=1$ it is true (see (*))

$$\mathcal{Z}\{\Delta^{n+1} f_k\} = (z-1)\,\mathcal{Z}\{\Delta^n f_k\} - z\Delta^n f_0 =$$

$$= (z-1)^{n+1} F(z) - z\sum_{j=0}^{n-1} (z-1)^{n-j} \Delta^j f_0 - z\Delta^n f_0 =$$

$$= (z-1)^{n+1} F(z) - z\sum_{j=0}^{n} (z-1)^{n-j} \Delta^j f_0 \qquad \square$$

Difference equations

(in czech diferenční rovnice)

The difference equation can be written as

$$\alpha_n \Delta^n y_k + \alpha_{n-1} \Delta^{n-1} y_k + \dots + \alpha_1 \Delta^1 y_k + \alpha_0 y_k = f_k ,$$

where $\{y_k\}$ is unknown sequence

$\{f_k\}$ is given sequence

and α_j are constants (which depends e.g. on sampling frequency)

This difference equation can be rewritten as recursive formula (and vice versa)

$$\beta_n y_{k+n} + \beta_{n-1} y_{k+n-1} + \dots + \beta_1 y_{k+1} + \beta_0 y_k = f_k$$

Solution of difference equations by z-transform

Example: Let's consider equation

$\Delta^2 y_k - y_k = f_k$ with initial conditions $y_0 = 0$, $\Delta^1 y_0 = 0$

and $f_0 = 0$, $f_k = (-1)^{k-1} - 1$, $k > 0$

i.e. $\{f_k\} = \{0, 0, -2, 0, -2, 0, -2, \cdots\}$

$$\mathcal{Z} \Big| \quad \Delta^2 y_k - y_k = f_k$$

$$(z-1)^2 Y(z) - Y(z) = F(z),$$

where $Y(z) = \mathcal{Z}\{y_k\}$

$$F(z) = \mathcal{Z}\{f_k\} = \sum_{k=1}^{\infty} \frac{-2}{z^{2k}} = \sum_{k=0}^{\infty} \frac{-2}{z^{2k}} + 2 =$$

$$= -2 \sum_{k=0}^{\infty} \left(\frac{1}{z^2}\right)^k + 2 = \frac{-2}{1 - \frac{1}{z^2}} + 2 =$$

$$= \frac{-2z^2}{z^2 - 1} + 2 = \frac{-2}{z^2 - 1} = \frac{2}{1 - z^2}$$

$$(z-1)^2 Y(z) - Y(z) = \frac{2}{1 - z^2}$$

$$Y(z) = \frac{-2}{z(z+1)(z-1)(z-2)}$$

$$\{y_k\} = \mathcal{Z}^{-1}\{Y(z)\}$$

93

1) inverse transform "by hand":

$$\frac{-2}{z(z+1)(z-2)(z-1)} = \frac{A}{z} + \frac{B}{z+1} + \frac{C}{z-2} + \frac{D}{z-1}$$

$\vdots$

$A = \dots, \quad B = \dots, \quad C = \dots, \quad D = \dots$

i.e. $\{y_k\} = \mathcal{Z}^{-1}\left\{\frac{A}{z}\right\} + \mathcal{Z}^{-1}\left\{\frac{B}{z+1}\right\} + \mathcal{Z}^{-1}\left\{\frac{C}{z-2}\right\} +$

$$+ \mathcal{Z}^{-1}\left\{\frac{D}{z-1}\right\}$$

where $\frac{A}{z}$ = "$0 + \frac{A}{z} + \frac{0}{z^2} + \frac{0}{z^3} + \dots$"

$$\Rightarrow \mathcal{Z}^{-1}\left\{\frac{A}{z}\right\} = \{0, A, 0, 0, \dots\} = \{A\delta_1\}$$

$$\frac{B}{z+1} = B\,\frac{1}{z}\,\frac{1}{1+\frac{1}{z}} = \frac{B}{z}\,\frac{1}{1-\left(-\frac{1}{z}\right)} =$$

$$= \frac{B}{z}\sum_{k=0}^{\infty}\left(-\frac{1}{z}\right)^k = B\sum_{k=0}^{\infty}\frac{(-1)^k}{z^{k+1}} =$$

$$= -B\sum_{k=0}^{\infty}\frac{(-1)^{k+1}}{z^{k+1}} = -B\left(\sum_{k=0}^{\infty}\frac{(-1)^k}{z^k} - 1\right)$$

$$\Rightarrow \mathcal{Z}^{-1}\left\{\frac{B}{z+1}\right\} = \left\{-B(-1)^k - B\delta_0\right\}$$

$\mathcal{Z}^{-1}\left\{\frac{C}{z-2}\right\}$ and $\mathcal{Z}^{-1}\left\{\frac{D}{z-1}\right\}$ are computed analogously

2) inverse transform by Maple

```
with(inttrans):
invztrans(-2/z/(z+1)/(z-1)/(z-2), z, k);
```

$$\rightarrow \quad -\operatorname{charfcn}_1(k) - \frac{1}{2}\operatorname{charfcn}_0(k) - \frac{1}{3}(-1)^k - \frac{1}{6}2^k + 1 = y_k$$

where $\operatorname{charfcn}_1(k) = \delta_1$
$\operatorname{charfcn}_0(k) = \delta_0$

<u>Another example:</u> Fibonacci sequence

$1, 1, 2, 3, 5, 8, 13, \dots$ which is given by recursive formula $a_{k+2} = a_{k+1} + a_k$,

$a_0 = a_1 = 1$.

$$\mathcal{Z} \Big| \quad a_{k+2} = a_{k+1} + a_k$$

$$z^2 A(z) - z^2 a_0 - z a_1 = z A(z) - z a_0 + A(z)$$

see the Maple file

95

```
> with(inttrans):
> rov:=a(k+2)=a(k+1)+a(k);
```

$$rov := a(k+2) = a(k+1) + a(k)$$

```
> a(0):=1;
```

$$a(0) := 1$$

```
> a(1):=1;
```

$$a(1) := 1$$

```
> alias(A(z)=ztrans(a(k),k,z)):
> ROV:=ztrans(rov,k,z);
```

$$ROV := z^2 A(z) - z^2 - z = z\, A(z) - z + A(z)$$

```
> RES:=solve(ROV,A(z));
```

$$RES := \frac{z^2}{z^2 - z - 1}$$

```
> RES1:=convert(RES,parfrac);
```

$$RES1 := 1 + \frac{1+z}{z^2 - z - 1}$$

```
> res:=invztrans(RES1,z,k);
```

$$res := \frac{1}{5} \sum_{\_\alpha = \mathrm{RootOf}(-1 + \_Z + \_Z^2)} \frac{(1 + 2\,\_\alpha)\left(\frac{1}{\_\alpha}\right)^k}{\_\alpha}$$

```
> seq([k,res],k=1..8);
```

[1, 1], [2, 2], [3, 3], [4, 5], [5, 8], [6, 13], [7, 21], [8, 34]

```
> seq([k,res],k=100..100);
```

[100, 573147844013817084101]

```
>
```

Rem.: using z-transform we can get e.g. a_{100} without computing the whole sequence "$a_0, a_1, \dots, a_{99}, a_{100}$"

Another example fully solved by Maple:

consider difference equation

$$\Delta^2 f_n = f_n + n^2$$

and initial data $f_0 = 1, \ f_1 = -1$

---

for solution see Maple file

97

```
> restart:with(inttrans):with(LREtools):
> alias(F(z)=ztrans(f(n),n,z)):
> rov:=delta(f(n),n,2)=f(n)+n*n;
```

$$rov := f(n+2) - 2\, f(n+1) + f(n) = f(n) + n^2$$

```
> ROV:=ztrans(rov,n,z);
```

$$ROV := z^2\, F(z) - f(0)\, z^2 - f(1)\, z - 2\, z\, F(z) + 2\, f(0)\, z + F(z) = F(z) + \frac{z\,(z+1)}{(z-1)^3}$$

```
> f(0):=1;f(1):=-1;
```

$$f(0) := 1$$

$$f(1) := -1$$

```
> RES:=solve(ROV,F(z));
```

$$RES := \frac{z^4 - 6\, z^3 + 12\, z^2 - 9\, z + 4}{z^4 - 5\, z^3 + 9\, z^2 - 7\, z + 2}$$

```
> res:=invztrans(RES,z,n);
```

$$res := 2\, charfcn_0(n) - n^2 - 2 + 2^n$$

```
> p:=seq([n,res],n=0..6);
```

$$p := [0, 1], [1, -1], [2, -2], [3, -3], [4, -2], [5, 5], [6, 26]$$

```
> with(plots):
Warning, the name changecoords has been redefined
> pointplot({p},symbol=circle,color=red);
```



Example (population model):

We consider ecosystem with vegetation (grass, ...) wolves and rabbits. The amount of each species is given by

$$
\begin{aligned}
v &= \bar{v} + v_K && \text{... vegetation} \\
w &= \bar{w} + w_K && \text{... wolves} \\
r &= \underset{\substack{\uparrow \\ \text{average}}}{\bar{r}} + \underset{\substack{\uparrow \\ \text{perturbation in K-th year}}}{r_K} && \text{... rabbits}
\end{aligned}
$$

There is a proposition for linear perturbation model of population

$$
\begin{pmatrix} v_{K+1} \\ w_{K+1} \\ r_{K+1} \end{pmatrix} = \begin{pmatrix} 1{,}5 & 0 & -0{,}25 \\ 0 & 0{,}75 & 0{,}25 \\ 4 & -8 & 0{,}5 \end{pmatrix} \begin{pmatrix} v_K \\ w_K \\ r_K \end{pmatrix} \qquad (M)
$$

We would like to know the evolution of rabbit population, i.e. $\{r_K\} = ?$ for initial data $r_0 = 0{,}77$, $r_1 = 2{,}11$, $r_2 = 3{,}78$

(99)

First we need to derive a recursive formula for $\{r_k\}$ from model (M) eliminating $\{v_k\}$ and $\{w_k\}$. We define the shift operator S:

$$S a_k = a_{k+1}$$

$$S(S a_k) = S^2 a_k = a_{k+2}$$

$$\vdots$$

We can write (M) as

$$\begin{pmatrix} S-1{,}5 & 0 & 0{,}25 \\ 0 & S-0{,}75 & -0{,}25 \\ -4 & 8 & S-0{,}5 \end{pmatrix} \begin{pmatrix} v_k \\ w_k \\ r_k \end{pmatrix} = \begin{pmatrix} 0 \\ 0 \\ 0 \end{pmatrix}$$

Let's multiply the 1-st equation by 4 and the 3-rd eq. by $(S-1{,}5)$

$$\begin{pmatrix} 4(S-1{,}5) & 0 & 1 \\ 0 & S-0{,}75 & -0{,}25 \\ -4(S-1{,}5) & 8(S-1{,}5) & (S-1{,}5)(S-0{,}5) \end{pmatrix} \begin{pmatrix} v_k \\ w_k \\ r_k \end{pmatrix} = \begin{pmatrix} 0 \\ 0 \\ 0 \end{pmatrix}$$

Now let's add the 1-st eq. to the 3-rd one

(100)

$$\begin{pmatrix} 4(s-1{,}5) & 0 & 1 \\ 0 & s-0{,}75 & -0{,}25 \\ & 8(s-1{,}5) & (s-1{,}5)(s-0{,}5)+1 \end{pmatrix} \begin{pmatrix} v_k \\ w_k \\ r_k \end{pmatrix} = \begin{pmatrix} 0 \\ 0 \\ 0 \end{pmatrix}$$

Now let's multiply the 2-nd equation by $-8(s-1{,}5)$ and the 3-rd one by $(s-0{,}75)$ and sum up both of them we get

$$\left[(s-0{,}75)\left((s-1{,}5)(s-0{,}5)+1\right)+2(s-1{,}5)\right] r_k = 0$$

$$\left[(s-0{,}75)\left(s^2-2s+1{,}75\right)+2s-3\right] r_k = 0$$

$$\left(s^3-2{,}75s^2+5{,}25s-4{,}3125\right) r_k = 0$$

$\Rightarrow$ we get recursive formula

$$r_{k+3}-2{,}75\,r_{k+2}+5{,}25\,r_{k+1}-4{,}3125\,r_k=0$$

with $r_0 = 0{,}77$, $r_1 = 2{,}11$, $r_2 = 3{,}78$,

which is solved by z-transform,

see the Maple file

(101)

```
> restart:
> with(inttrans):
> rov:=r(k+3)-2.75*r(k+2)+5.25*r(k+1)-4.3125*r(k)=0;
```

$$rov := r(k+3) - 2.75\, r(k+2) + 5.25\, r(k+1) - 4.3125\, r(k) = 0$$

```
> r(0):=0.77;
```

$$r(0) := 0.77$$

```
> r(1):=2.11;
```

$$r(1) := 2.11$$

```
> r(2):=3.78;
```

$$r(2) := 3.78$$

```
> alias(R(z)=ztrans(r(k),k,z)):
> ROV:=ztrans(rov,k,z);
```

$$ROV := z^3\, R(z) - 0.77\, z^3 + 0.0075\, z^2 - 2.0200\, z - 2.75\, z^2\, R(z) + 5.25\, z\, R(z) - 4.3125\, R(z) = 0$$

```
> RES:=solve(ROV,R(z));
```

$$RES := \frac{0.04000000000\, z\, (308.\, z^2 - 3.\, z + 808.)}{16.\, z^3 - 44.\, z^2 + 84.\, z - 69.}$$

```
> RES1:=convert(RES,parfrac);
```

$$RES1 := 0.7700000000 + \frac{1.340390199}{z - 1.280303308} + \frac{0.9328010156 + 0.7696098006\, z}{z^2 - 1.469696692\, z + 3.368342464}$$

```
> res:=invztrans(RES1,z,k);
```

$$res := 1.046931763\; 1.280303308^k - 0.1384658813\, (0.7348483459 - 1.681767039\, I)^k + 0.2893125582\, I\, (0.7348483459 - 1.681767039\, I)^k - 0.1384658813\, (0.7348483459 + 1.681767039\, I)^k - 0.2893125582\, I\, (0.7348483459 + 1.681767039\, I)^k$$

```
> seq([k,Re(res)],k=0..6);
```

[0, 0.7700000004], [1, 2.109999999], [2, 3.780000000], [3, 2.638124999], [4, -3.490781262], [5, -7.148554706], [6, 10.04499024]

```
>
```

Sequence $\{r_k\}$ exhibits growing oscillations, it means that either ecosystem is unstable or the model for population is wrong.

102

<u>Remarks</u> Recursive formula for linear perturbation model

$$\begin{pmatrix} v_{k+1} \\ w_{k+1} \\ r_{k+1} \end{pmatrix} = \begin{pmatrix} 1{,}5 & 0 & -0{,}25 \\ 0 & 0{,}75 & 0{,}25 \\ 4 & -8 & 0{,}5 \end{pmatrix} \begin{pmatrix} v_k \\ w_k \\ r_k \end{pmatrix}$$

is in fact iteration method

$X_{k+1} = U X_k$ which is convergent ("has stable behavior") if $\rho(U) \leq 1$, where $\rho(U) = \max\limits_i |\lambda_i|$ is spectral radius of matrix U (λ_i are eigenvalues of U).

Let's try to compute eigenvalues of U by Maple

103

```
> restart:with(LinearAlgebra):
> U:=Matrix([[1.5,0,-0.25], [0,0.75,0.25], [4,-8,0.5]]);
```

$$U := \begin{bmatrix} 1.5 & 0 & -0.25 \\ 0 & 0.75 & 0.25 \\ 4 & -8 & 0.5 \end{bmatrix}$$

```
> L:=Eigenvalues(U);
```

$$L := \begin{bmatrix} 1.28030330849110908 + 0.\,I \\ 0.734848345754444576 + 1.68176703860097064\,I \\ 0.734848345754444576 - 1.68176703860097064\,I \end{bmatrix}$$

```
> abs(L[1]);
```

$$1.280303308$$

```
> abs(L[2]);
```

$$1.835304461$$

```
> abs(L[3]);
```

$$1.835304461$$

```
>
```

$\Rightarrow \quad \rho(U) = 1{,}835304461 > 1$

$\Rightarrow$ iteration method has growing oscillations (is unstable).

Example of discrete Filter:

Let's consider analog RC filter



with input voltage $x(t)$ and output voltage $y(t)$. The filter can be desribed by two equations

$$R\,i(t) + y(t) = x(t)$$
$$i(t) = y'(t)\cdot C$$

Eliminating the current $i(t)$ we get one ODE

$$\underline{RC\,y'(t) + y(t) = x(t)}$$

Now we want to create a discrete RC filter.

Let's discretize variable t: $t_K = K\cdot h$, $h>0$

and $y(t_K) = y_K$, $x(t_K) = x_K$.

a) first version of discrete filter (implicit)

$$y'(t_{K+1}) = \frac{y_{K+1} - y_K}{h} + O(h), \text{ i.e.}$$

$$RC\,\frac{y_{K+1} - y_K}{h} + y_{K+1} = x_{K+1}$$

105

(A) $\underbrace{y_{k+1} - b \cdot y_k = d \cdot x_{k+1}}$ discrete RC filter

with $b = \frac{RC}{RC+h}$, $d = \frac{h}{RC+h}$

z-transform of (A):

$$z\,Y(z) - z y_0 - b\,Y(z) = d\,z\,X(z) - d\,z\,x_0$$

let's consider $x_0 = 0$, $y_0 = 0$, then

$$Y(z) = \underbrace{\frac{dz}{z-b}}_{H(z)}\,X(z)$$

$H(z)$ ... transfer function

$$\{y_k\} = \{h_k\} * \{x_k\}, \text{ where}$$

$$\{h_k\} = \mathcal{Z}^{-1}\{H(z)\}$$

$$H(z) = d\,\frac{1}{1-\frac{b}{z}} = d \sum_{k=0}^{\infty} \left(\frac{b}{z}\right)^k \Rightarrow h_k = d \cdot b^k$$

To have sequence $\{h_k\} * \{x_k\}$ bounded we necessarily need $|b| < 1$, i.e.

$\left|\frac{RC}{RC+h}\right| < 1$, which is satisfied for any $h \Rightarrow$ (A) is stable realization of RC filter

(106)

b) let's try another realization of RC filter (explicit)

$$y'(t_k) = \frac{y_{k+1} - y_k}{h} + O(h) \quad , \text{ i.e.}$$

$$RC \, \frac{y_{k+1} - y_k}{h} + y_k = x_k$$

(B) $\underbrace{y_{k+1} - b\, y_k = d\, x_k}$ also discrete RC filter now with $b = \frac{RC - h}{RC}$, $d = \frac{h}{RC}$

z-transform of (B):

$$z\, Y(z) - z y_0 - b\, Y(z) = d\, X(z)$$

again consider $y_0 = 0$

$$Y(z) = \underbrace{\frac{d}{z-b}}_{H(z)} X(z)$$

now $$H(z) = d \, \frac{1}{z} \, \frac{1}{1 - \frac{b}{z}} = \frac{d}{z} \sum_{k=0}^{\infty} \frac{b^k}{z^k} = \frac{d}{b} \sum_{k=0}^{\infty} \frac{b^{k+1}}{z^{k+1}} =$$

$$= \frac{d}{b} \sum_{k=0}^{\infty} \frac{b^k}{z^k} - \frac{d}{b}$$

$$\Rightarrow \{h_k\} = \{0, d, db, db^2, db^3, \dots\}$$

107

we again need $|b|<1$, i.e.

$$\left|\frac{RC-h}{RC}\right|<1 \Rightarrow h \in (0; 2RC)$$

So the realization (B) is stable for

$h \in (0; 2RC)$.